

**Escola de Engenharia Elétrica, Mecânica e de Computação
Universidade Federal de Goiás**



Laboratório de Microprocessadores e Microcontroladores

Experimento 2: Interrupções Externas e Relés

Alunos: _____ Matrícula: _____

Prof. Dr. José Wilson Lima Nerys

Goiânia, 1º semestre de 2019

SUMÁRIO

1	<i>Interrupções Externas e Relés</i>	3
1.1	Interrupções Externas.....	3
1.2	Relés.....	5
2	<i>Tarefas do Experimento 2.....</i>	6
2.1	Tarefa 1 – Rotação de LEDs com interrupção	6
2.2	Tarefa 2 – Rotação de LEDs com prioridade alta da interrupção 1	7
2.3	Tarefa 3 – LEDs piscando com interrupção	8
2.4	Tarefa 4 – Acionamento de Lâmpada de 220 V através de Relé	8
2.5	Tarefa 5 – Acionamento Temporizado de Lâmpada de 220 V	9
2.6	Tarefa 6 – Revisão de Algumas Instruções	10

1 Interrupções Externas e Relés

1.1 Interrupções Externas

Interrupção é o processo pelo qual a execução de um programa é interrompida para a execução de um outro processamento que pode ser solicitado por uma das três fontes abaixo:

Interrupção por software (instrução)
 Interrupção pedida por periférico externo
 Interrupção pedida por periférico interno (temporizador/contador e porta serial)

O microcontrolador 8051 pode ser interrompido de **cinco** maneiras diferentes:

- Pela interrupção externa INTO\ - pino 12 (P3.2)
- Pelo timer/counter (temporizador/contador) interno TIMER0
- Pela interrupção externa INT1\ - pino 13 (P3.3)
- Pelo timer/counter (temporizador/contador) interno TIMER1
- Pelo canal de comunicação serial (Pinos 10 e 11 = P3.0 e P3.1)

O pedido de interrupção pode ou não ser atendido, de acordo com a condição de certos registradores. No 8051 os registradores que comandam a interrupção são

- Registrador de Habilitação: **IE** (*Interrupt Enable*) = Reg. A8h
- Registrador de Prioridades: **IP** (*Interrupt Priority*) = Reg. B8h
- Registrador de Controlador: **TCON** (*Timer Control*) = Reg. 88h

Registrador IE : (Reg. A8h)	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
	EA	x	x	ES	ET1	EX1	ET0	EX0

EA (*Enable All*) - Quando está zerado (EA = 0), todas as interrupções estão desabilitadas (mascaradas), independentemente de seus bits individuais de controle. Quando está setada (EA = 1), cada uma das interrupções pode ser habilitada ou desabilitada fazendo seus bits de controle 1 ou 0.

EX0 (*Enable External Interrupt 0*) - Quando está zerado (EX0 = 0) a interrupção externa, cujo pedido vem através do pino INTO\ está desabilitada. Quando está setado (EX0 = 1), a interrupção INTO\ fica habilitada.

ET0 (*Enable Timer 0*) - Quando ET0 = 0, a interrupção pedida pelo temporizador/contador 0 fica desabilitada. Quando ET0 = 1, a interrupção vinda do temporizador/contador 0 fica habilitada.

EX1 (*Enable External Interrupt 1*) - Quando está zerado (EX1 = 0) a interrupção externa, cujo pedido vem através do pino INT1\ está desabilitada. Quando está setado (EX1 = 1), a interrupção INT1\ fica habilitada.

ET1 (*Enable Timer 1*) - Quando ET1 = 0, a interrupção pedida pelo temporizador/contador 1 fica desabilitada. Quando ET1 = 1, a interrupção vinda do temporizador/contador 1 fica habilitada.

ES (*Enable Serial*) - Quando ES = 0, a interrupção pedida pela porta serial fica desabilitada. Quando ES = 1 essa interrupção fica habilitada.

Registrador IP : (Reg. B8h)	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
	x	x	x	PS	PT1	PX1	PT0	PX0

PX0 (*Priority of External Interrupt 0*) - Quando $PX0 = 1$ a interrupção externa $INT0\backslash$ recebe prioridade alta.

PT0 (*Priority of Timer/Counter Interrupt 0*) - Quando $PT0 = 1$ a interrupção pedida pelo temporizador/contador 0 recebe prioridade alta.

PX1 (*Priority of External Interrupt 1*) - Quando $PX1 = 1$ a interrupção externa $INT1\backslash$ recebe prioridade alta.

PT1 (*Priority of Timer/Counter Interrupt 1*) - Quando $PT1 = 1$ a interrupção pedida pelo temporizador/contador 1 recebe prioridade alta.

PS (*Priority of Serial Port Interrupt*) - Quando $PS = 1$ a interrupção pedida através da porta serial recebe prioridade alta.

Quando $PX0$, $PT0$, $PX1$, $PT1$ e PS são zero, elas são de prioridade baixa. Caso uma interrupção de prioridade 1 seja solicitada durante a execução de uma de prioridade 0, o processamento é interrompido para o atendimento da interrupção de prioridade maior.

No caso de todas as interrupções terem a mesma prioridade (0 ou 1), a ordem de atendimento das interrupções é:

- Interrupção externa 0 —————> Maior prioridade
- Temporizador/contador 0
- Interrupção externa 1
- Temporizador/contador 1
- Canal serial —————> Menor prioridade

As interrupções externas podem ser ajustadas para serem detectadas por nível 0 ou pela transição do nível 1 para o nível 0. O ajuste é feito através do registrador **TCON**, dado abaixo.

Registrador TCON : (Reg. 88h)	Controle do Temporizador				Bit 3	Bit 2	Bit 1	Bit 0
	TF1	TR1	TF0	TR0	IE1	IT1	IE0	IT0

IT0 (*Interrupt 0 Type*) - Quando $IT0 = 1$ a interrupção externa 0 será reconhecida pela transição de 1 para 0 no pino $INT0\backslash$. Quando $IT0 = 0$, a interrupção é reconhecida quando o sinal no pino $INT0\backslash$ está em nível baixo (0).

IE0 (*Interrupt 0 Edge Flag*) - É setado pelo hardware quando uma interrupção externa através de $INT0\backslash$ é detectada. É zerada quando da execução da instrução **RETI** (retorno da subrotina de atendimento).

IT1 (*Interrupt 1 Type*) - Quando $IT1 = 1$ a interrupção externa 1 será reconhecida pela transição de 1 para 0 no pino $INT1\backslash$. Quando $IT1 = 0$, a interrupção é reconhecida quando o sinal no pino $INT1\backslash$ está em nível baixo (0).

IE1 (*Interrupt 1 Edge Flag*) - É setado pelo hardware quando uma interrupção externa através de $INT1\backslash$ é detectada. É zerada quando da execução da instrução **RETI** (retorno da subrotina de atendimento).

Endereços de desvio das interrupções

Quando ocorre uma das cinco interrupções do 8051 o processamento é desviado para os endereços mostrados na Tabela 1. Como há pouco espaço em bytes nesses endereços, deve-se usar uma instrução de desvio para outro endereço, onde seja possível escrever toda a rotina de atendimento da interrupção, caso a rotina de atendimento seja maior que 8 bytes.

Tabela 1: Endereços das interrupções

Interrupção Solicitada	Endereço de desvio
Reset	0000h
INT0\	0003h
Timer/counter 0	000Bh
INT1\	0013h
Timer/counter 1	001Bh
Canal Serial	0023h

1.2 Relés

A Fig. 1 mostra os circuitos de acionamento de lâmpadas que fazem parte do módulo do Relé do Kit Didático. Em um dos circuitos um transistor BC548 é usado para o acionamento do relé (com bobina de 12 V), através do qual uma lâmpada de 12 V é acionada. Outro circuito é utilizado para acionamento de uma lâmpada de 220 V, com a utilização de um relé com bobina de 5 V e um mosfet BS170D.




Fig. 1 - Circuito de acionamento das lâmpadas de 12 V e 220 V

A Fig. 2 mostra o circuito do sensor de presença, que é composto por um LED infravermelho e um fototransistor. A montagem do LED e do fototransistor é tal que, na presença de um obstáculo, a luz do LED reflete no obstáculo e incide na base do fototransistor, levando o nível de saída (P3.3) de alto (5 V) para baixo (0 V). Assim, a interrupção externa 1 (Pino P3.3) pode ser usada para o acionamento de uma das lâmpadas, através do sensor de presença.




Fig. 2 - Circuito do sensor de presença

2 Tarefas do Experimento 2

Os programas das tarefas a seguir devem ser **digitados e compilados no MCU 8051** e **executados** no simulador do kit didático e no kit didático do microcontrolador 8051.

2.1 Tarefa 1 – Rotação de LEDs com interrupção

A Tabela 2 apresenta um programa onde a interrupção externa 0 é usada para rotacionar para a esquerda os LEDs da porta P1 e a interrupção externa 1 é usada para rotacionar os LEDs para a direita.

Tabela 2: Rotação de LEDs usando as interrupções externas

Rótulo	Mnemônico	Comentário sobre o Efeito da Operação
	ORG 00H	
	LJMP INICIO	
	ORG 03H	
	LJMP ESQUERDA	; Desvia para a subrotina de atendimento da interrupção INT0
	ORG 13H	
	LJMP DIREITA	; Desvia para a subrotina de atendimento da interrupção INT1
	ORG 30H	
INICIO:	MOV SP, #2FH	
	MOV IE, #85H	; Habilita interrupções externas 0 e 1 (EA = 1, EX0 = 1 e EX1 = 1)
	MOV TCON, #04H	; INT0 por nível (IT0 = 0) e INT1 por transição de 1 para 0 (IT1 = 1).
	MOV A, #01H	; Valor inicial do acumulador
	SJMP \$; Aguarda interrupção em um laço infinito
ESQUERDA:	MOV P1,A	; Transfere o conteúdo do acumulador para a porta de saída P1
	RL A	; Rotaciona para a esquerda o conteúdo de A
	LCALL ATRASO	; Chama subrotina de atraso de tempo
	RETI	; retorna de subrotina de interrupção
DIREITA:	MOV R7,#16	; R7 recebe o valor decimal 16
V2:	RR A	; Rotaciona para a direita o conteúdo do acumulador
	MOV P1,A	; Transfere o conteúdo do acumulador para a porta P1
	LCALL ATRASO	; Chama subrotina de atraso de tempo
	DJNZ R7,V2	; Decrementa o conteúdo de R7; enquanto não for zero, volta para V2
	CLR IE1	; Elimina interrupção externa 1 pendente
	RETI	; Retorna de subrotina de interrupção
ATRASO:	CLR EA	; Limpa o bit EA do registrador IE. Desabilita todas as interrupções
	MOV R0,#3	; Faz R0 = 3 decimal
V3:	MOV R1,#200	; Faz R1 = 200 decimal
V4:	MOV R2,#250	; Faz R2 = 250 decimal
	DJNZ R2, \$; Decrementa R2 até zerar (desvia pra ele mesmo)
	DJNZ R1, V4	; Decrementa R1. Se R1 não for zero, desvia para V4
	DJNZ R0,V3	; Decrementa R0. Se R0 não for zero, desvia para V3

	SETB EA	; Seta o bit EA do registrador IE. Habilita novamente as interrupções
	RET	; Retorna da subrotina de atraso de tempo
	END	

Compile o programa da Tabela 2. Execute o programa usando o **simulador do Kit Didático** e observe o funcionamento do programa quando as interrupções zero e 1 são solicitadas. Verifique o funcionamento do programa também no **Kit Real**.

Questão 1: Qual é a diferença no funcionamento para a ESQUERDA e para a DIREITA?

2.2 Tarefa 2 – Rotação de LEDs com prioridade alta da interrupção 1

No programa original da Tabela 2 acrescente a instrução **MOV IP,#04H** após a instrução **MOV TCON,#04H**. Compile o programa e o **execute no Simulador do Kit Didático** e no **Kit Real**. No registrador especial **IP**, mostrado a seguir, verifica-se que IP = 04H significa prioridade alta para a interrupção externa 1 (PX1 = 1).

IP	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
B8H	x	x	x	PS	PT1	PX1	PT0	PX0

IP=04H	0	0	0	0	0	1	0	0
--------	---	---	---	---	---	---	---	---

Avalie a seguinte condição de funcionamento:

Mantenha a chave da interrupção 0 (P3.2) pressionada e, durante a rotação dos LEDs para a esquerda, acione uma vez a chave da interrupção 1 (P3.3).

Questão 2: O que ocorre? Explique.

2.3 Tarefa 3 – LEDs piscando com interrupção

Escreva, na Tabela 3, as subrotinas CASO1 e CASO2, que devem substituir as subrotinas ESQUERDA e DIREITA do programa da Tabela 2. Substitua a instrução MOV TCON,#04H por MOV TCON,#05H. As novas subrotinas devem atender a duas situações:

Situação 1 (CASO 1): quando a interrupção INT0 for acionada, os LEDs devem piscar 10 vezes (10 vezes ligados e 10 vezes desligados).

Situação 2 (CASO 2): quando a interrupção INT1 for acionada, os LEDs ímpares e os pares devem piscar alternadamente 10 vezes.

Tabela 3: Adaptação do programa da Tabela 2

Rótulo	Mnemônico
CASO1:	
	RETI

Rótulo	Mnemônico
CASO2:	
	RETI

2.4 Tarefa 4 – Acionamento de Lâmpada de 220 V através de Relé

A Tabela 4 apresenta um programa para acionamento de uma lâmpada de 220 V através da interrupção externa 0.

Tabela 4: Acionamento de Lâmpada de 220 V através da interrupção externa 0

Rótulo	Mnemônico
	LAMP EQU P3.4
	ORG 00H
	LJMP INICIO
	ORG 03H
	CPL LAMP
	CLR IE0
	RETI

Rótulo	Mnemônico
	ORG 30H
INICIO:	MOV SP, #2FH
	MOV IE, #81H
	MOV TCON, #01H
	SJMP \$
	END

Questão 3: Como funciona o programa da Tabela 4?

2.5 Tarefa 5 – Acionamento Temporizado de Lâmpada de 220 V

A Tabela 5 apresenta um programa para acionamento temporizado de uma lâmpada de 220 V, através de um sensor de presença, conectado ao pino da interrupção externa 1 (Pino P3.3). Na presença de um obstáculo a lâmpada acende e, após 10 s (aproximadamente), ela apaga. A lâmpada está conectada ao pino P3.4. Verifique o funcionamento no **Kit Didático Real**.

Tabela 5: Acionamento temporizado de uma lâmpada de 220 V através da interrupção externa 1

Rótulo	Mnemônico	Rótulo	Mnemônico
	LAMP EQU P3.4	V1:	SETB LAMP
			MOV R7,#10
	ORG 00H	V2:	LCALL ATRASO1s
	LJMP INICIO		DJNZ R7,V2
			CLR LAMP
	ORG 13H		CLR IE1
	LJMP V1		RETI
	ORG 30H	ATRASO1s:	MOV R0,#20
INICIO:	MOV SP, #2FH	V3:	MOV TH0,#HIGH(19455)
	MOV TMOD,#01H		MOV TL0,#LOW(19455)
	MOV IE, #84H		SETB TR0
	MOV TCON, #04H		JNB TF0,\$
	CLR LAMP		CLR TF0
	SJMP \$		DJNZ R0,V3
			CLR TR0
			RET
			END

Questão 4: Como funciona o programa da Tabela 5?

2.6 Tarefa 6 – Revisão de Algumas Instruções

Instrução	Função
MOV IE, #81H	
MOV IE, #85H	
MOV TCON, #04H	
MOV TCON, #05H	
MOV IP, #04H	
SJMP \$	
RL A	
RR A	
CLR EA	
SETB EA	
SETB LAMP	
CLR LAMP	
CLR IE0	
CLR IE1	
DJNZ R0, V3	
JNB TF0, \$	